
Count cells equal to one of many things

Generic formula 

= SUMPRODUCT(COUNTIF(rng,things))

Summary 

To count the number of cells equal to one of many values, you can use the COUNTIF function
inside of the SUMPRODUCT function. In the generic form of the formula above, rng represents a
range of cells, and things represents the values to count. In the example shown, cell G5 contains
this formula:

= SUMPRODUCT(COUNTIF(B5:B10,things))

where things is the named range E5:E7.

Note: COUNTIF is not case-sensitive.

Explanation 

In this example, the goal is to count the values in column B listed in the range E5:E7. One way to
do this is to give the COUNTIF function all three values in the named range things (E5:E7) as
criteria, then use the SUMPRODUCT function to get a total. The formula in G4 is:

= SUMPRODUCT(COUNTIF(B5:B15,things))

The COUNTIF function counts the number of cells in a range that meet criteria. When you give
COUNTIF a range of cells as the criteria, it returns an array of numbers as the result, where each
number represents the count of one thing in the criteria range. In this case, the named range
things (D5:D7) contains 3 values, so COUNTIF returns 3 results in an array as shown below:

= COUNTIF(B5:B15,things) 
= COUNTIF(B5:B15,{"apples";"pears";"kiwis"}) 
= {2;3;1} // result from COUNTIF

Since "apple" appears twice, "pears" appears three times, and "kiwis" appears once, the array
contains the numbers 2, 3, and 1. This array is returned directly to the SUMPRODUCT function:

= SUMPRODUCT({2;3;1})

With a single array to process, SUMPRODUCT all items in the array and returns a final result, 6.

With an array constant

With a limited number of values, you can use an array constant in your formula like this:

= SUMPRODUCT(COUNTIF(B5:B15,{"apples","pears","kiwis"}))

ISNUMBER and MATCH

The above formula works fine, but has some limitations due to the nature of COUNTIF. As an
alternative, you can use the formula below, which uses the ISNUMBER function with the MATCH
function to achieve the same result:

= SUMPRODUCT( -- ISNUMBER(MATCH(B5:B15,things,0)))

This is a more flexible formula in cases where logical conditions become more complex. It's also
useful when you need to extract a value from a range in the data to use in a condition.

Author 

Dave Bruns

Related formulas 

Summary count with COUNTIF
When working with data, a common need is to perform summary
calculations that show total counts in different ways. For
example, total counts by category, color, size, status, etc. The
COUNTIF function is a good way to generate these kinds of
totals...

Count cells equal to either x or y
COUNTIF counts the number of cells in a range that match
supplied criteria. In this case, the criteria for the first COUNTIF is
"apples" and the criteria for the second COUNTIF is "pears". The
first COUNTIF returns the count of cells in B5:B10 equal...

Count total matches in two ranges
Suppose you have a "master" list of some kind, and also have
another list that contains some of the same items. You want a
formula that compares the values in the 2nd list to see how many
of them appear in the first list. You don't care about the...

Sum if equal to one of many things
In this example, the goal is to sum the numbers in column C
when the item in column B is listed in the range E5:E7. One way
to accomplish this is to give the SUMIF function all three values
in the named range things (E5:E7) as criteria, then use the...

Related functions 

Excel COUNTIF Function
COUNTIF is an Excel function to count cells in a range that meet a
single condition. COUNTIF can be used to count cells that contain
dates, numbers, and text. The criteria used in COUNTIF supports
logical...

Excel SUMPRODUCT Function
The Excel SUMPRODUCT function multiplies ranges or arrays
together and returns the sum of products. This sounds boring,
but SUMPRODUCT is an incredibly versatile function that can
be...

Excel ISNUMBER Function
The Excel ISNUMBER function returns TRUE when a cell contains
a number, and FALSE if not. You can use ISNUMBER to check that
a cell contains a numeric value, or that the result of another
function is a number.

Excel MATCH Function
MATCH is an Excel function used to locate the position of a
lookup value in a row, column, or table. MATCH supports
approximate and exact matching, and wildcards (* ?) for partial
matches. Often, MATCH is combined...

See also 

Excel's RACON functions

Excel Formula Training
Formulas are the key to getting things done in
Excel. In this accelerated training, you'll learn how to
use formulas to manipulate text, work with dates and
times, lookup values with VLOOKUP and INDEX &
MATCH, count and sum with criteria, dynamically
rank values, and create dynamic ranges. You'll also

I just wanted to let you know that after years of being afraid of formula (we never learned Excel at
school) and relying on others to do the complicated stuff - i found your website and it is now my go-

approach, clarity and logic when breaking down a formula and how it could be used. -Anna

